NBS

Technical Note 242

A FORTRAN PROGRAM FOR ANALYSIS OF ELLIPSOMETER MEASUREMENTS AND CALCULATION OF REFLECTION COEFFICIENTS FROM THIN FILMS

FRANK L. McCRACKIN
AND JAMES P. COLSON

U. S. DEPARTMENT OF COMMERCE
NATIONAL BUREAU OF STANDARDS

THE NATIONAL BUREAU OF STANDARDS

The National Bureau of Standards is a principal focal point in the Federal Government for assuring maximum application of the physical and engineering sciences to the advancement of technology in industry and commerce. Its responsibilities include development and maintenance of the national standards of measurement, and the provisions of means for making measurements consistent with those standards; determination of physical constants and properties of materials; development of methods for testing materials, mechanisms, and structures, and making such tests as may be necessary, particularly for government agencies; cooperation in the establishment of standard practices for incorporation in codes and specifications; advisory service to government agencies on scientific and technical problems; invention and development of devices to serve special needs of the Government; assistance to industry, business, and consumers in the development and acceptance of commercial standards and simplified trade practice recommendations; administration of programs in cooperation with United States business groups and standards organizations for the development of international standards of practice; and maintenance of a clearinghouse for the collection and dissemination of scientific, technical, and engineering information. The scope of the Bureau's activities is suggested in the following listing of its four Institutes and their organizational units.

Institute for Basic Standards. Electricity. Metrology. Heat. Radiation Physics. Mechanics. Applied Mathematics. Atomic Physics. Physical Chemistry. Laboratory Astrophysics.* Radio Standards Laboratory: Radio Standards Physics; Radio Standards Engineering.** Office of Standard Reference Data.

Institute for Materials Research. Analytical Chemistry. Polymers. Metallurgy. Inorganic Materials. Reactor Radiations. Cryogenics.** Office of Standard Reference Materials.

Central Radio Propagation Laboratory.** Ionosphere Research and Propagation. Troposphere and Space Telecommunications. Radio Systems. Upper Atmosphere and Space Physics.

Institute for Applied Technology. Textiles and Apparel Technology Center. Building Research. Industrial Equipment. Information Technology. Performance Test Development. Instrumentation. Transport Systems. Office of Technical Services. Office of Weights and Measures. Office of Engineering Standards. Office of Industrial Services.

---

* NBS Group, Joint Institute for Laboratory Astrophysics at the University of Colorado.

** Located at Boulder, Colorado.

NATIONAL BUREAU OF STANDARDS

Technical Note 242

ISSUED MAY 27, 1964

A FORTRAN PROGRAM FOR ANALYSIS OF ELLIPSOMETER MEASUREMENTS AND CALCULATION OF REFLECTION COEFFICIENTS FROM THIN FILMS

Frank L. McCrackin

and James P. Colson

NBS Technical Notes are designed to supplement the Bureau's regular publications program. They provide a means for making available scientific data that are of transient or limited interest. Technical Notes may be listed or referred to in the open literature.

For sale by the Superintendent of Documents, U.S. Government Printing Office
Washington, D.C., 20402 - Price 30 cents

Contents

A Fortran Program for Analysis of Ellipsometer Measurements and Calculation of Reflection Coefficients from Thin Films

Frank L. McCrackin

A Fortran computer program to calculate the reflection coefficients for both single and multiple thin films and to analyze optical measurements of such films by an ellipsometer is presented. Both the films and the substrate may be absorbing and have complex indexes of refraction. The reflection coefficients of inhomogeneous films (films of varying refractive index) may be computed by determining the reflection coefficients of an equivalent series of homogeneous films.

The ellipsometer measurements for a given film may be calculated or the index of refraction and thickness of a film may be calculated from ellipsometer readings. The ellipsometer measurements may be corrected for the retardation of the compensator or for multiple reflections of the light.

1. Introduction

The reflection of light from surfaces and thin films has been discussed by various authors [1,2]. The reflection is expressed in terms of the reflection coefficients R^p for light polarized parallel to the plane of incidence and R^s for light polarized perpendicular to the plane of incidence. Since these coefficients are complex quantities, they represent both the change of amplitude and phase of the light on reflection.

The ellipsometer is an optical instrument used to study surfaces and their films by reflection of light [3,4]. The ellipsometer measures the ratio of the reflection coefficients

$$\rho = \frac{R^p}{R^s} = \tan \psi e^{i\Delta}$$

The ellipsometer contains a polarizer, analyzer, and compensator. For the commonly occurring case of the relative retardation Q of the compensator equal to 90°, and the arrangement of components given by reference 3, ρ is related to the settings of the polarizer P and the analyzer A by

$$P = \Delta/2 - 45°$$

$$A = \psi$$

The ratio of reflection coefficients, ρ, is a function of the refractive index of the substrate, n_3, (which may be complex) the thickness d_j, and refractive index n_j of any of j films on the surface, some or all of which may be complex, the refractive index of the surrounding medium, n_1, the angle of incidence, φ_1, the wavelength of the light used, and the retardation of the quarter wave plate. Hence the ellipsometer readings P and A are related to the optical constants of a surface and the thickness and index of refraction of films on the surface. However, since the calculations required to determine these quantities from ellipsometer measurements (and also the inverse calculations) are very time consuming, this Fortran program was consequently developed for this purpose.

The following computations can be performed by this program:

1) Computation of ellipsometer readings and reflection coefficients for given refractive indexes of surrounding medium, film and substrate and given film thickness.

2) Computation of film thickness from ellipsometer readings for given refractive indexes of film and substrate

3) Computation of film thickness and refractive index of film from ellipsometer readings for given refractive index of substrate.

4) Computation of refractive index of a bare substrate from ellipsometer readings.

2. Methods of Calculation

Most of the formulas used in this program are given in McCrackin et al [3]. The correction of ellipsometer readings for a compensator which has a retardation other than quarter wave, is given by Winterbottom [2]. The calculations involved in use of multiple reflections of the light from several surfaces as well as discussion of increase of sensitivity and experimental technique is given by Stromberg et al [5]. The calculation from P and A of the thickness of a film of given refractive index (Instructions No. 903) has also been described by McCrackin et al [3].

I will now describe briefly the manner of introducing data and instructions into the computer and some of the calculations which may be performed by this program; a

detailed discussion of the program will be found in the Section entitled "Computer Instructions".

The first card for any calculation is a title card that contains any desired identifying information. Following cards start with an instruction number in columns 1 to 3 of the cards, followed by numbers starting in columns 10, 20, 30, 40, 50, and 60. All numbers except the instruction number must contain a decimal point. The instruction number determines the meaning of subsequent numbers and the calculation (if any) to be performed. Instruction numbers from 001 to 901 are used to introduce data; those from 902 to 907 instruct the computer to perform a certain calculation.

For a substrate without a film, the ratio of reflection coefficients, ρ, depends on the angle of incidence, ϕ, of the light, the refractive index, n_1, of the surrounding medium, and the refractive index, n_3, of the substrate. The refractive index of the substrate, n_3, may be calculated from the measured value of ρ (given by values of P and A), ϕ and n_1. This calculation is done by the instruction number 905 after ϕ and n_1 have been given by a 001 instruction.

For a film on a substrate, ellipsometer readings P and A and the reflection coefficients may be computed from ϕ, n_1, n_3, and the thickness, d_2, and refractive index, n_2, of the film. This calculation is performed by the 902 instruction after n_1 and ϕ are given by an 001 instruction, n_2 is given by an 002 instruction, and n_3 is given by an 003 instruction. The thickness d_2 is given by the 902 instruction. Alternatively, the.thickness of a film, d_2, may be computed from P, A, ϕ, n_1, n_2, and n_3 by instruction 903. The P and A values are given by the 903 instruction after ϕ and n_1 are given by an 001 instruction, n_2 is given by an 002 instruction, and n_3 by an 003 instruction.

For a non-absorbing film, with real index of refraction, both d_2 and n_2 may be computed by the 904 instruction. The P and A values are given by the 904 instruction after ϕ and n_1 are given by an 001 instruction, n_3 by an 003 instruction, and a range of values that includes the solution for n_2 by the 002 instruction.

The required instructions to perform some typical calculations for single films are now described. The Section entitled "Computer Instructions" gives detailed instructions for the similar calculations for multiple films as well as the other calculations that may be performed by this program.

2.1 Calculation of Ellipsometry Readings and Reflection Coefficients for a Film Covered Surface

Instruction number 902 will cause the computer to calculate the ellipsometer readings and reflection coefficients for reflection of light from a film-covered surface. The thickness of the film is given with the 902 instruction card. Before the 902 instruction, instructions 001, 002, and 003 must be used to give the other values required for the calculation. The 001 instruction gives the index of refraction of the medium above the film, the angle of incidence of the light, and the wavelength of the light. The 002 instruction gives the index of refraction of the film and the 003 instruction gives the index of refraction of the substrate. Also, every calculation must be terminated by a 910 instruction. For example, let light of wavelength 4800 Å be incident at 65° in air on a film covered surface. Let the thickness of the film be 40 Å, the refractive index of the film be 1.4, and the complex refractive index of the substrate be 3.316-4.383 i. The following input data computes the ellipsometer reading, P and A, and the reflection coefficients, R^p and R^s:

Column No. 1	10	20	30	
JONES TEST CALCULATION				3-10-64
001	1.	65.	4800.	
002	1.4			
003	(blank)	3.316	4.383	
902	40.			
910				

Notice that the absolute value of the imaginary component of the refractive index is used.

Tables of ellipsometer readings and reflection coefficients for films of different thicknesses and refractive indexes may also be computed by the 902 instruction. In this case, the initial value, increment, and final value to be used are given for the thickness and refractive index of the film. Thus, the input data:

Column No. 1	10	20	30	
JONES TEST CALCULATION				3-10-64
001	1.	65.	4800.	
002	1.4	0.1	1.6	
003	(blank)	3.316	4.383	
902	0.	20.	100.	
910				

will produce tables for films of refractive index of 1.4, 1.5, and 1.6 for each table, the ellipsometer readings and reflection coefficients will be given for films 0., 20., 40., 60., 80., and 100 Å thick.

Calculations may also be performed for absorbing films of complex refractive index, as described in the section entitled, "Computer Instructions".

2.2 Calculation of Index of Refraction for Film-Free Substrate from Ellipsometer Readings

Instruction 905 calculates the index of refraction of a film-free surface from ellipsometer readings P and A. The index is printed and may also be used in the following calculations.

2.3 Calculation of Film Thickness from Ellipsometer Readings for Given Index of Refraction of Film

Instruction 903 computes the thickness of a film from ellipsometer readings P and A. The 001 and 002 instruction must precede this instruction and the index of the substrate must have been either given by a 003 instruction or calculated by a 905 instruction. For example, suppose a film-free surface was measured under water (refractive index of 1.33) to give P = 27.32° and A = 30.45°. Then an absorbing film of refractive index 2.5 - 0.3 i was placed on the surface and reading of P = 31.15 and A = 29.00° were obtained. Let the angle of incidence be 65° and the wavelength of the light be 4800 Å in vacuum. The following input data will calculate the thickness of the film:

Column No.

1	10	20	30	40
JONES CAL FILM THICKNESS				
001	1.33	65.	4800.	
905	27.32	30.45		
002	2.5	(blank)	(blank)	.3
903	31.15	29.00		
910				

If the 002 instruction specifies a series of values for the refractive index of the film, a film thickness can be computed using each value of the series for the refractive index of the film.

These examples illustrate the manner in which the input data is written. Instructions are also available to do calculations for multiple films, to calculate the index of refraction of adsorbed films, to determine ranges of error of the calculated refractive index of a film, to correct the ellipsometer readings for reflection of light from multiple surfaces, and to correct the ellipsometer readings for a compensator of arbitrary relative retardation. A detailed explanation of all the instruction numbers, a list of subroutines used by the program, a list of variables used in the program, the fortran program, and a flow diagram of the program follow.

3. Computer Instructions

This program can perform calculations for reflection of light from one to 898 films. The case of five films is shown in Figure 1.

The refractive indices are labelled n_1 to n_7 and the thicknesses of the films are d_2 to d_6. The index n_1 of the medium is assumed real, but all the other indices may be complex. The angle of incidence of the light in the medium is ϕ, and is measured in degrees. The vacuum wavelength of

the light is λ. The film thicknesses and λ must be expressed in the same units. The relative retardation of the quarter wave plate in the ellipsometer is Q, measured in degrees, and is assumed to be 90° unless otherwise specified.

The first line of the data is a title identifying the run. Each following line (or card) of data consists of an instruction number followed by the data. Each instruction number determines the meaning of subsequent data and the calculation to be performed. The instructions are as follows:

Instruction No. 001

This instruction gives values of n_1, ϕ, λ, α, number of reflections, and Q for use in later calculations. n_1 is the refractive index of the surrounding medium, ϕ is the angle of incidence, λ is the wave length of the light in vacuum, α is a number used to control the calculations for instruction No. 904 and will be discussed under instruction 904, and Q is the relative retardation of the compensator used in the ellipsometer. The instruction reads as follows:

Column No.

1	10	20	30	40	50	60
001	n_1	ϕ	λ	α	No reflections	Q

The computer assumes the initial values

$n_1 = 1$

$\phi = 70°$

$\lambda = 5461.$

$\alpha = 0$

No. of Reflections = 1

$Q = 90°$

and if these values are to be used, the instruction 001 does not need to be given. Also, values need be given only when they are to be changed. Thus, to give the values $n_1 = 1.5$, $\phi = 65°$, $\lambda = 5461$, $\alpha = 0$, number of reflections = 5, and $Q = 90°$, the instruction would be

Columns

1	10	20	30	40	50	60
001	1.5	65.			5.	

If later calculations were to be performed for the same values except that ϕ is changed to 75°, the required 001 instruction is

Columns

1	10	20
001		75.

Notice that, except for instruction numbers, all numbers given as input data must have a decimal point.

Instruction No. 002

This instruction gives values of the index of refraction, n_2, of the top film for use in following calculations. If only real values of n_2 are to be used, the arrangement of the 002 card is

Columns 1	10	20	30
002	Initial n_2	Increment	Final n_2

For example, to specify $n_2 = 3$, 3.5 and 4, the instruction is:

Columns 1	10	20	30
002	3.	.5	4.

To specify complex values of the refractive index, the values to be used for the imaginary component are specified after the real component. For example, the card:

Columns 1	10	20	30	40	50	60
002	3.	.5	4.	0.	.1	.2

will cause all following calculations to be done for these values of n_2:

3.	3.5	4
3-0.1 i	3.5-0.1 i	4.-0.1 i
3-0.2 i	3.5-0.2 i	4 -0.2 i

A single value for the refractive index may be given by omitting the increment and final value for the real and imaginary component. Thus the value $n_2 = 3-i$ is specified by:

Columns 1	10	20	30	40	50	60
002	3.			1.		

Instructions No. 003 to 900

These instructions give the thicknesses and indices of the films and substrate. Thus, if film number 4 was 100 Å thick with an index of 1.4 - 2.2 i, the line of data would be

004 100. 1.4 2.2

No thickness is given for the substrate. An instruction for a film may be repeated. For example, for five films with medium and substrate lines with instruction numbers 001 to 007 would be given. These would be followed by instructions causing calculations. Then the thickness and index of any film could be changed by repeating its instruction number and additional calculations could be performed with the original values for the other films. For more than one film, the last of these cards must contain 1. starting in column 40 on the card.

Instruction No. 901

This instruction gives values of ρ_1, ρ_p, n_p, $\delta' P$, and $\delta' A$ to be used in subsequent calculations. The values ρ_1, ρ_p and n_p are used to compute the mass of material per unit area of the film for instructions 903 and 904. The values $\delta' P$ and $\delta' A$ are used by instruction 904. If this instruction is used, all values must be given.

Instruction No. 902

This instruction causes the computer to calculate tables of P, A, and reflection coefficients for the films. The

range of thicknesses of the upper film are given on the line. Thus

902   0.   5.   15.

will cause the calculation for d_2 = 0., 5., 10, and 15 Å and for all values of n_2 given previously.

The values of d_2 may be omitted from the 902 instruction if they have been previously given. For example, a section of data may be

001   1.3

902   0.   10.   100.

001   1.4

902

001   1.5

902   100.   10.   200.

910

Tables of P and A will be printed for d_2 = 0 to 100 in steps of 10 for n_1 = 1.3 and 1.4, and for d = 100 to 200 in steps of 10 for n_1 = 1.5.

Instruction No. 903

In this case, readings P and A are given. The computer calculates and prints a value of d_2 to give the best fit [3] to the P and A readings for every n_2 given by the 002 instruction. It also computes the value of P and A corresponding exactly to the n_2 and d_2 values and prints the differences δP and δA between these computed P and A

values and the P and A readings.

Two solutions for d_2 and a measure of error for each solution are computed by the method given in reference 3. These solutions are obtained as solutions of a quadratic equation by taking the plus or minus sign for the radical. The solution with the smallest measure of error, labeled THICK, with its measure of error, labeled ERROR, and δP and δA, labeled DEL P and DEL A, are printed on the left hand side of the printout, and the sign of the radical for this solution is printed to the extreme right of the printout. The other solution for the thickness with corresponding values of δP, δA and measure of error are printed on the right hand side of the printout.

If the film is not compact but is a mixture of the surrounding medium and a material of density ρ_p and refractive index n_p, the mass of material per unit area of the film is computed in mg/cm^2 and is printed. The value of ρ_1, density of surrounding medium g/cm^3, ρ_p in g/cm^3 and n_p are given by the 901 instruction; the Lorentz-Lorentz equation is used to compute the refractive index of the film.

If compensator with retardation other than 90° is used, values of P, a_p and a_s must be given (listed in this order) instead of merely P and A [2, 3].

Instruction No. 904

The measured values of P and A are given. The computer determines the film index, n_2, and film thickness corresponding to the values of P and A, if a solution for n_2 exists in the range of values given for n_2 by the previous 002 instruction. If $n_2 = n_{2r} - in_{2i}$ is complex and a range of values of n_{2i} has been specified by a 002 instruction, a value of n_{2r} will be computed for each of the values of n_{2i} specified.

The computer finds the solution for n_{2r} by computing the film thickness for assumed values of n_{2r} [3]. This computed thickness is a complex number. The real part of the computed thickness is chosen as the film thickness and the imaginary part, d_i, as a measure of error. Then the P and A values are calculated for this film index and real thickness and their differences δP, δA from the given values of P and A are computed. These differences are then compared with the allowed differences $\delta' P$ and $\delta' A$ given by the 901 instruction. If the differences δP and δA are both smaller (in absolute value) than the allowed differences $\delta' P$ and $\delta' A$, the computer prints the results of the calculation for this value of n_{2r}; otherwise a new value of n_{2r} is chosen and the calculation repeated until δP and δA are smaller than $\delta' P$ and $\delta' A$.

If values of $\delta' P$ and $\delta' A$ have not been given by a 901 instruction, the computer assumes $\delta' P = \delta' A = .0005°$.

In particular, the film thickness and error terms δP, δA and d_i are first computed for the initial and final values of n_{2r} given by the previous 002 instruction. The two error terms d_i for the two values of n_{2r} are examined. Since d_i must be zero for the correct value of n_{2r}, it is expected to change sign as n_{2r} varies from the initial to final value of the specified range. Therefore, if the d_i values for the initial and final values of n_{2r} have the <u>same</u> sign, the computer assumes that there is no solution within the specified range of n_{2r}. The information printed for this case is determined by the sign of α as given by a previous 001 instruction. If it is not specified or α is positive, no printout is given. However, if α has a negative value, the comment "No solution in range of n_2" is printed and the results of the calculation for the initial and final values of n_{2r} are printed. If the range of values for n_{2r} specified by the 002 instruction is too large, two or more solutions for n_{2r} may occur in the range of n_{2r}, so the error term d_i may have the same sign at each end of the range of n_{2r}. In this case, the computer will erroneously assume that there is no solution in the range of n_{2r}. In practice, this has only occurred when n_2 is complex.

However, if d_i has different signs for the initial and final values of n_{2r}, the computer performs the calculations for a value of n_{2r} equal to the average of the two previous values of n_{2r}. If the error terms δP and δA for this solution are less than $\delta' P$ and $\delta' A$, the computer prints the solution. Otherwise, it determines in which half of the range of n_2 the solution lies. If the signs of d_i for initial, average, and final values of the refractive index of the film are + -- or -++, the solution for n_2 lies between the initial and average values, otherwise the solution lies between the average and final values. Using this reduced range for n_2 the procedure is repeated until the error terms δP and δA are less than $\delta' P$ and $\delta' A$ in absolute value. The procedure will be repeated only 20 times as a precautionary measure against use of excessive computation time.

Thus an index of refraction n_2 and thickness of a film is computed from measured values of P and A. This value of n_2 is the best value determined from the measured P and A values but will contain an error due to errors of measurement of the P and A values. The computer will determine the limits of error for n_{2r} from the limits of error of P and A if a value for σ has been previously given. The upper limit of error is determined first. The increment of n_{2r} (given by preceding 002 instruction) is added to the

best value of n_2 and film thickness and error terms computed for this value. If the error term in either P (δP) or A(δA) is larger than the limit of experimental error of P and A, this value of n_{2r} is taken as the upper limit for n_{2r}. Otherwise, the increment of n_{2r} is again added and the process repeated. This process is repeated until one of three things occur:

(1) Either error term is larger than the experimental error of P or A.

(2) The value of n_{2r} is larger than the upper limit for n_{2r} (given by 002 instruction).

(3) The process has been performed α times. (The absolute value of α is taken, as α may be negative).

The value of α is, of course, given by the 001 instruction. A value of 20 for α has been found to be satisfactory.

The lower limit for n_2 is determined similarly with the increment of n_2 subtracted instead of added. In this case, the calculation stops if the value of n_2 is less than the lower limit for n_2.

If a compensator with a retardation different from 90° is used in the ellipsometer, values of p, a_p, and a_s (listed in this order) instead of P and A must be given with this instruction.

Instruction No. 905

This instruction calculates the index of the substrate from P and A readings of the bare substrate. The index of the substrate may be given by an instruction 003 to 900 or computed by this instruction. If the retardation of the compensator is different than 90°, values of p, a_p, and a_s must be given.

Instruction No. 906

The experimental errors of P and A are given. These are used for instruction 904 when α has been given.

Instruction No. 907

Experimental values of p, a_p, and a_s are given and the value Q of the compensator is calculated. The notation and equations are given by Winterbottom [2].

Instruction No. 908

This instruction allows another title card to follow it.

Instruction No. 909

This instruction sets all the values that have been given by instructions 001, 902, and 906 to their initial values. It is followed by a title card.

The number of films may be increased by giving instructions 003 to 900, but may not be decreased. Suppose, for example, the system of 5 films previously shown (with refractive indexes n_2 to n_7) has been given to the computer. Then to describe a system of 7 films (indexes n_1 to n_9)

instructions 008 and 009 with instructions for any other films that are to be changed in thickness of index are all that are required. However, to describe a system of less than 5 films, instruction 909 must be used.

Instruction No. 910

This instruction stops the calculation. It must be the last instruction of every set of data.

To perform calculations involving multiple reflection of the light, or where an imperfect quarter wave plate ($Q \neq 90°$) has been used, the number of reflections or value of Q are given by 001 instruction. For an imperfect quarter wave plate, values of P, a_p, and a_s are given with instructions 903, 904, and 905, also the value of Q may also be computed by a 907 instruction.

Every set of data must start with a title card and instructions 908 and 909 must be followed by a title card. A title card may contain any information desired in columns 1 to 80.

The program is written in Fortran II for use on IBM machines 704, 7090 or 7094 with the Bell Monitor. For use with other Monitors, the READ, PRINT, and CALL SYSTEM statements would have to be changed.

4. Functions and Subroutines for Ellipsometry Program

ABSF	Absolute value
SQRTF	Square root
SINF	Sine of argument in radians
COSF	Cosine of argument in radians
SINDF	Sine of argument in degrees
COSDF	Cosine of argument in degrees
LOGF	Logarithm to base e
ATANDF	Arctangent, result in degrees
ATN1F(Y,X)	Arctangent of Y/X with result in proper quadrant in radians
SYSTEM	End of computation
CSR	Complex square root, square root of a complex no.
CD	Complex division
CM	Complex multiplication
CS(C3,X,FN3)	Cosine of angle of propagation of light

$$C3 = \sqrt{1 - (n_1 \sin \phi_1/n_3)^2}$$

$X = n_1 \sin \phi_1$, $FN3 = n_3$; n_3 and C 3 complex numbers

SRC(R, T1, T2) Small reflection coefficient

R, T1, T2 complex

$$R = \frac{T1 - T2}{T1 + T2}$$

RC(CR, RP21, RP32, Q) Capital reflection coefficient

CR, RP21, RP32, and Q are complex

$P = 4\pi Q$

$$CR = \frac{RP21 + RP32\ e^{iP}}{1 + RP21 * RP32\ e^{-iP}}$$

$$Q = \frac{n_2 \cos \phi_2 \ d_2}{\lambda}$$

5. Variables Used in Fortran Program

Dimensional Variables

B(2)	Coefficient in equation for y
C(2)	Coefficient in equation for y
C2(2)	Cos ϕ_2
C3(3)	Cos ϕ_3
CI(2)	Cos ϕ_i
CIP1(2)	Cos ϕ_{n+1}
CN(2)	Cos ϕ_n
CNM1(2)	Cos ϕ_{n-1}
CRN(2)	Total reflection coefficient (normal polarization)
CRN21(2)	Total reflection coefficient between films 2 and 1 (normal polarization)
CRN32(2)	Total reflection coefficient between films 3 and 2 (normal polarization)
CRP(2)	Total reflection coefficient (parallel polarization)
CRP21(2)	Total reflection coefficient between films 2 and 1 (parallel polarization)
CRP32(2)	Total reflection coefficient between films 3 and 2 (parallel polarization)
D(900)	d_L - - - - thickness of film L
D2C(2)	d_2 - - - - computed from y_1
D2CX(2)	d_2 - - - - computed from y_2

DFN2(2)	Increment for film refractive index
E(2)	Coefficient in equation for y
FN2(2)	n_2 - - - - refractive index of upper film
FN2E(2)	Final value for film refractive index
FN2I(2)	Initial value for film refractive index
FN2X(2)	n_2 - - - - with imaginary part positive
FNX(900,2)	n_L - - - - refractive indexes of film $_L$
IDT(12)	Title
RHO(2)	$\rho = \dfrac{R^p{}_{21}}{R^n_{21}} = \tan \psi \epsilon^{i\Delta}$
RN(2)	$r^n{}_{n-1,n}$ reflection coefficient between substrate and adjacent film (normal polarization)
RN21(2)	$r^n{}_{21}$ - - - - reflection coefficient between films 2 and 1 (normal polarization)
RN32(2)	$r^n{}_{32}$ - - - - reflection coefficient between films 3 and 2 (normal polarization)
RP(2)	$r^p{}_{n-1,n}$ reflection coefficient between substrate and adjacent film (parallel polarization)
RP21(2)	$r^p{}_{21}$ - - - - reflection coefficient between films 2 and 1 (parallel polarization)
RP32(2)	$r^p{}_{32}$ - - - - reflection coefficient between films 3 and 2 (parallel polarization)
T1(2) to T5(2)	Temporary storage for complex numbers
XX(2)	Ln(-y)
Y1(2) Y2(2)	Solutions for $y = \epsilon^{-4\pi_i n_2 \cos\phi_2 d_2/\lambda}$ of equation $Ey^2 + By + C = 0$

NON-DIMENSIONAL VARIABLES

A	A	Angle of analyzer
ADX	$A_{cal}-A_{exp}$	Difference between calculated and experimental values of A
AP	A_p	
AS	A_s	
AX		Calculated A for single and multiple reflections
AXX	Measured A	
C1	Cos ϕ_1	
DC	A	Adsorption of polymer
DD2	Increment of d_2	
DEL	$= 2P + 90°$	Relative phase retardation
DENP	ρ_p	Density of polymer
DEN1	ρ_1	Density of solvent
D1	d_2	Computed film thickness
D2	d_2	Thickness of upper film
D2F	Final d_2	
D2I	Initial d_2	
EA EP		Experimental errors in A and P, used for calculating allowed range of values for film thickness
ERA ERP	ΔA ΔP	Allowed errors in A and P for calculating n_2
ER1 ER2 ER3		Errors in d_2 for real part of $n_2 = X_1$, X_2, and X_3
FN		Number of reflections
FNP	n_p	Index of polymer

FN1	n_1	Refractive index of surrounding mediu
HOLD		Temporary storage
I		Index and n-2
J		Index
L		Instruction numbers
M	L-900	Reduced instruction numbers
MX		Index for computed go to
MY		Index for computed go to
N		Number of layers
NX	N-4	
P	P	Angle of polarizer
PDX	$P_{cal}-P_{exp}$	Difference between calculated and experimental values of p
PX	P	Calculated p for single and multiple reflections
PXX	P	Corrected for quarter wave plate
PHI1	O_1	Angle of incidence
QWP	Q or δK	Retardation of compensator
S1	Sin ϕ_1	
THETA1	θ_1	Phase of R^p_{21}
Theta2	θ_2	Phase of R^n_{21}
TNA	Tan A	
TNPSI	Tan ψ	
WL	λ	Vacuum wavelength of light
WP		Used in computing DC
X	$n_1 \mathrm{Sin}\phi_1$	
XM1		Magnitude of R^p_{21}

XM2		Magnitude of $R^n{}_{21}$
X1 X2 X3		Trial values of real part of n_2
XXL	1	Winterbottom
Z	+ or -	Sign used in solution of quadratic eq.
Z1 to Z6		Temporary storage for input data

6. FORTRAN PROGRAM

```
      DIMENSION CRN(2),CN(2),RN(2),RP(2),FNX(2,900),D(900),XX(2)
      DIMENSION FN2(2),FN2I(2),DFN2(2),FN2E(2),FN2X(2),C2(2),C3(2)
      DIMENSION RP21(2),RN21(2),RP32(2),RN32(2),Y1(2),Y2(2),E(2),B(2)
      DIMENSION C(2),T1(2),T2(2),T3(2),T4(2),T5(2),RHO(2),IDT(14),CRP(2)
      DIMENSION D2C(2),D2CX(2),CRN32(2),CRP32(2),CRN21(2),CRP21(2)
      DIMENSION Q(2),CNM1(2),CI(2),CIP1(2)
   10 FN1=1.
      PHI1=70.
      WL=5461.
      DEN1=1.
      DENP=1.
      FNP=1.5
      N= 3
      ERP=.0005
      ERA=.0005
      D2I=0.
      DD2=0.
      D2F=0.
      NCH=0.
      QWP=0.
      FN=1.
      EP=.006
      EA=.02
   20 READ 30,IDT
   30 FORMAT(13A6,A2)
      PRINT 40,IDT
   40 FORMAT(1H113A6,A2,38H    PROG. 10 MULTIPLE FILM ELLIPSOMETRY)
   50 READ 60,L,Z1,Z2,Z3,Z4,Z5,Z6,Z7,Z8
   60 FORMAT(I3,6X,6F10.0,A6,A5)
      PRINT 70,L,Z1,Z2,Z3,Z4,Z5,Z6,Z7,Z8
   70 FORMAT(1H0I3,6F12.5,3X,A6,A5)
      IF(L-1)80,170,80
   80 IF(L-2)90,290,90
   90 IF(L-900)120,120,100
  100 M=L-900
      IF(M-9)101,101,550
  101 GO TO(110,1050,1100,1100,1100,2070,670,20,10),M
  110 DEN1= Z1
      DENP= Z2
      FNP= Z3
      ERP= Z4
      ERA= Z5
      GO TO 50
  120 D(L)=Z1
      FNX(1,L)= Z2
      FNX(2,L)= -ABSF(Z3)
      N= XMAXOF(N,L)
      IF(Z4)130,50,130
  130 X= S1*FN1
      CALL CS(CN,X,FNX(1,N))
      CALL CS(CNM1,X,FNX(1,N-1))
      CALL CM(T1,FNX(1,N),CNM1)
      CALL CM(T2,FNX(1,N-1),CN)
```

```
      CALL SRC(CRP,T1,T2)
      CALL CM(T1,FNX(1,N-1),CNM1)
      CALL CM(T2,FNX(1,N),CN)
      CALL SRC(CRN,T1,T2)
      IF(N-4)160,160,140
  140 NX= N-4
      I= N-2
      DO 150 J=1,NX
      CALL CS(CIP1,X,FNX(1,I+1))
      CALL CS(CI,X,FNX(1,I))
      CALL CM(T1,FNX(1,I+1),CI)
      CALL CM(T2,FNX(1,I),CIP1)
      CALL SRC(RP,T1,T2)
      CALL CM(T1,FNX(1,I),CI)
      CALL CM(T2,FNX(1,I+1),CIP1)
      CALL SRC(RN,T1,T2)
      Q(1)=T2(1)*D(I+1)/WL
      Q(2)=T2(2)*D(I+1)/WL
      CALL RC(CRN,RN,CRN,Q)
      CALL RC(CRP,RP,CRP,Q)
  150 I= I-1
  160 GO TO 50
  170 IF(Z1)180,190,180
  180 FN1=Z1
  190 IF(Z2)200,210,200
  200 PHI1= Z2
  210 IF(Z3)220,230,220
  220 WL= Z3
  230 IF(Z4)240,250,240
  240 NCH=Z4
  250 S1= SINDF(PHI1)
      C1= COSDF(PHI1)
      IF(Z5)260,270,260
  260 FN= Z5
  270 IF(Z6) 280,275,280
  275 QWP=Z6
  280 IF(N-4) 50,130,130
  290 FN2I(1)= ABSF(Z1)
      DFN2(1)= Z2
      FN2E(1)= Z3+DFN2(1)/2.
      FN2I(2)= ABSF(Z4)
      DFN2(2)= Z5
      FN2E(2)= Z6+DFN2(2)/2.
      GO TO 50
  320 IF(Z1)321,322,321
  321 D2I= Z1
  322 IF(Z2)323,324,323
  323 DD2= Z2
  324 IF(Z3)325,400,325
  325 D2F= Z3+DD2/2.
      GO TO 400
  380 I=N-2
      PRINT 390,I,FNX(1,N),FNX(2,N),FN,P,A
  390 FORMAT(1HOI3,23H FILMS,INDEX SUBSTRATE=2F9.5,2H I,F5.0,13H REFLECT
     1IONS.2X,2HP=F9.3,2X,2HA=F9.3/9HO    THICK3X,9HA MG/CMSQ4X,5HDEL P4X
```

```
   2,5HDEL A6X,5HERROR4X,5HTHICK4X,5HDEL P4X,5HDEL A6X,5HERROR3X,16HN2
   3 REAL  N2 IMAG)
400 FN2X(1)= FN2I(1)
    FN2X(2)= FN2I(2)
410 FN2(1) = FN2X(1)
    FN2(2) = -FN2X(2)
    MX= 1
420 X= FN1*S1
    CALL CS(C2,X,FN2)
    CALL CS(C3,X,FNX(1,3))
    T1(1)= FN2(1)*C1
    T1(2)= FN2(2)*C1
430 T2(1)= FN1*C2(1)
    T2(2)= FN1*C2(2)
    CALL SRC(RP21,T1,T2)
    CALL CM(T2,FN2,C2)
    T1(1)= FN1*C1
    T1(2)= 0.
    CALL SRC(RN21,T1,T2)
    IF(N-3)550,440,450
440 CALL CM(T1,FNX(1,3),C2)
    CALL CM(T2,FN2,C3)
    CALL SRC(CRP32,T1,T2)
    CALL CM(T1,FN2,C2)
    CALL CM(T2,FNX(1,3),C3)
    CALL SRC(CRN32,T1,T2)
    GO TO 470
450 CALL CM(T1,FNX(1,3),C2)
    CALL CM(T2,FN2,C3)
    CALL SRC(RP32,T1,T2)
    CALL CM(T1,FN2,C2)
    CALL CM(T2,FNX(1,3),C3)
460 CALL SRC(RN32,T1,T2)
    CALL CM(T1,FNX(1,3),C3)
    T2(1)= T1(1)*D(3)/WL
    T2(2)= T1(2)*D(3)/WL
    CALL RC(CRP32,RP32,CRP,T2)
    CALL RC(CRN32,RN32,CRN,T2)
470 GO TO(550,480,580,580),M
480 GO TO(481,483,485),MY
481 PRINT 482
482 FORMAT(1H06X,2HD26X,1HP8X,1HA9X,6HRF.CF.5X,1HP9X,6HRF.CF.5X,1HN7X,
   14HFN2P5X,4HFN2I)
    GO TO 486
483 PRINT 484
484 FORMAT(1H06X,2HD26X,1HP8X,2HAP8X,2HAS8X,6HRF.CF.5X,1HP9X,6HRF.CF.5
   1X,1HN7X,4HFN2R5X,4HFN2I)
    GO TO 486
485 PRINT 482
486 D2= D2I
    CALL CM(T1,FN2,C2)
500 T2(1)=T1(1)*D2/WL
    T2(2)=T1(2)*D2/WL
    CALL RC(CRP21,RP21,CRP32,T2)
    CALL RC(CRN21,RN21,CRN32,T2)
```

```
      CALL CD(RHO,CRP21,CRN21)
      PX=ATN1F(RHO(2),RHO(1))*28.64789-45.
      AX=ATANDF(SQRTF(RHO(1)**2+RHO(2)**2))
      XM1=SQRTF(CRP21(1)**2+CRP21(2)**2)
      THETA1=ATN1F(CRP21(2),CRP21(1))*57.2956
      XM2=SQRTF(CRN21(1)**2+CRN21(2)**2)
      THETA2=ATN1F(CRN21(2),CRN21(1))*57.2956
      GO TO(501,503,505),MY
  501 PRINT 502,D2,PX,AX,XM1,THETA1,XM2,THETA2,FN2(1),FN2(2)
  502 FORMAT(1H F9.1,2F9.3,F12.5,F9.3,F12.5,F9.3,2F9.4)
      GO TO 506
  503 PXX= .5*ATANDF(SINDF(GWP)*SINDF(2.*PX)/COSDF(2.PX))
      XL = .5*ACOSF(COSDF(2.*PXX)*COSDF(GWP))
      AP = ATANDF(SINDF(AX)*COSF(XL)/ COSDF(AX)/SINF(XL))
      AS = ATANDF(SINDF(AX)*SINF(XL)/COSDF(AX)/COSF(XL))
      PRINT 504,D2,PXX,AP,AS,XM1,THETA1,XM2,THETA2,FN2(1),FN2(2)
  504 FORMAT(1H F9.1,3F9.3,F12.5,F9.3,F12.5,F9.3,2F9.4)
      GO TO 506
  505 AXX= ATANDF(SINDF(AX)/COSDF(AX)**FN)
      PXX= (PX+45.)*FN-45.
      PRINT 502,D2,PXX,AXX,XM1,THETA1,XM2,THETA2,FN2(1),FN2(2)
  506 D2= D2+DD2
      IF(D2F-D2)520,520,500
  520 FN2X(2)= FN2X(2)+DFN2(2)
      IF(FN2F(2)-FN2X(2))530,530,410
  530 FN2X(1)= FN2X(1)+DFN2(1)
      FN2X(2)= FN21(2)
      PRINT 540
  540 FORMAT(1H )
      IF(FN2E(1)-FN2X(1))50,50,410
  550 CALL SYSTEM
  560 RHO(1)=SINDF(A)/COSDF(A)*COSDF(2.*P+90.)
      RHO(2)=SINDF(A)/COSDF(A)*SINDF(2.*P+90.)
      T1(1)=4.*S1**2*RHO(1)
      T1(2)=4.*S1**2*RHO(2)
      T2(1)=RHO(1)+1.
      T2(2)=RHO(2)
      CALL CM(T3,T2,T2)
      CALL CD(T2,T1,T3)
      T1(1)=1.-T2(1)
      T1(2)=-T2(2)
      CALL CSR(T2,T1)
      D(N)=0.
      FNX(1,N)= S1/C1*T2(1)*FN1
      FNX(2,N)= S1/C1*T2(2)*FN1
      PRINT 570,FNX(1,N),FNX(2,N),P,A
  570 FORMAT(20H0INDEX OF SUBSTRATE=F9.4,F9.4,2H I,2X,2HP=F9.3,2X,2HA=9.
     13)
      GO TO 130
  580 RHO(1)= SINDF(A)/COSDF(A)*COSDF(2.*P+90.)
      RHO(2)=SINDF(A)/COSDF(A)*SINDF(2.*P+90.)
      CALL CM(T1,CRN32,CRP32)
      CALL CM(T2,T1,RP21)
      CALL CM(T3,T2,RHO)
      CALL CM(T4,T1,RN21)
```

```
      E(1)=T3(1)-T4(1)
      E(2)=T3(2)-T4(2)
      CALL CM(T1,RN21,RP21)
      CALL CM(T2,T1,CRP32)
      T3(1)=CRN32(1)+T2(1)
      T3(2)=CRN32(2)+T2(2)
      CALL CM(T4,T3,RHO)
      CALL CM(T5,T1,CRN32)
      B(1)=T4(1)-CRP32(1)-T5(1)
      B(2)=T4(2)-CRP32(2)-T5(2)
      CALL CM(T1,RN21,RHO)
      C(1)=T1(1)-RP21(1)
      C(2)=T1(2)-RP21(2)
      CALL CM(T1,B,B)
      CALL CM(T2,E,C)
      T3(1)=T1(1)-4.*T2(1)
      T3(2)=T1(2)-4.*T2(2)
      CALL CSR(T4,T3)
      DO 590 IX= 1,2
      T1(IX)=-B(IX)+T4(IX)
      T2(IX)=-B(IX)-T4(IX)
  590 T3(IX)= 2.*E(IX)
      CALL CD(Y1,T1,T3)
      CALL CD(Y2,T2,T3)
      CALL CM(T1,FN2,C2)
      T3(1)=12.56637*T1(1)/WL
      T3(2)=12.56637*T1(2)/WL
      XX(1)=ATN1F(-Y1(2),Y1(1))
      XX(2)=0.5*LOGF(Y1(1)**2+Y1(2)**2)
      CALL CD(D2C,XX,T3)
      XX(1)=ATN1F(-Y2(2),Y2(1))
      XX(2)=0.5*LOGF(Y2(1)**2+Y2(2)**2)
      CALL CD(D2CX,XX,T3)
B     Z=606060606020
      IF(ABSF(D2C(2))-ABSF(D2CX(2)))610,610,600
  600 HOLD= D2C(1)
      D2C(1)=D2CX(1)
      D2CX(1)=HOLD
      HOLD=D2C(2)
      D2C(2)=D2CX(2)
      D2CX(2)=HOLD
B     Z=606060606040
  610 D1= D2C(1)
  620 T2(1)= T1(1)*D1/WL
      T2(2)=T1(2)*D1/WL
      CALL RC(CRP21,RP21,CRP32,T2)
      CALL RC(CRN21,RN21,CRN32,T2)
      CALL CD(RHO,CRP21,CRN21)
      WP=((FN1**2-1.)/(FN1**2+2.)-(FN2(1)**2-1.)/(FN2(1)**2+2.))*DENP/
     1 (((FN2(1)**2-1.)/(FN2(1)**2+2.))*(DEN1-DENP)-(FNP**2-1.)/
     2 (FNP**2+2.)*DEN1+(FN1**2-1.)/(FN1**2+2.)*DENP)
      DC=  WP/(WP/DENP+(1.-WP)/DEN1)*D1*1.E-5
      ADX=ATANDF(SQRTF(RHO(1)**2+RHO(2)**2))-A
      PDX=ATN1F(RHO(2),RHO(1))*28.64789-45.-P
      T2(1)=T1(1)*D2CX(1)/WL
```

```
      T2(2)=T1(2)*D2CX(1)/WL
      CALL RC(CRP21,RP21,CRP32,T2)
      CALL RC(CRN21,RN21,CRN32,T2)
      CALL CD(RHO,CRP21,CRN21)
      ADXX=ATANDF(SQRTF(RHO(1)**2+RHO(2)**2))-A
      PDXX=ATN1F(RHO(2),RHO(1))*28.64789-45.-P
      GO TO (625,820,830,880,1030,1190,2030),MX
  625 PRINT 630, D1,DC,PDX,ADX,D2C(2),D2CX(1),PDXX,ADXX,D2CX(2),FN2(1),
     1 FN2(2),Z
  630 FORMAT(1H F8.1,1PE12.4,0P2F9.3,F12.5,F8.1,2F9.3,F12.5,2F9.5,A6)
      GO TO 520
  670 P= Z1
      A=Z2
      AP=Z3
      TNPSI=SQRTF(SINDF(A)/COSDF(A)*SINDF(AP)/COSDF(AP))
      XL=ATN1F(TNPSI,SINDF(A)/COSDF(A))
      QWP=ACOSF(ABSF(COSF(2.*XL)/COSDF(2.*P)))*57.2956
      PRINT 680,QWP
  680 FORMAT( 45HORELATIVE RETARDATION OF QUARTER WAVE PLATE =F9.3)
      GO TO 50
  810 I=N-2
      PRINT 390,I,FNX(1,N),FNX(2,N),FN,P,A
      FN2X(2)=FN2I(2)
  315 FN2(2)=-FN2X(2)
      MX= 2
      X1= FN2I(1)
      X2= FN2E(1)
      FN2(1)= X2
      GO TO 420
  820 ER2= D2C(2)
      FN2(1)= X1
      MX= 3
      GO TO 420
  830 ER1= D2C(2)
      IF(ER1)840,840,850
  840 IF(ER2)855,855,870
  850 IF(ER2) 870,870,855
  855 IF(NCH)856,1031,1031
  856 PRINT 857
  857 FORMAT(27HONO SOLUTION IN RANGE OF N2)
  860 PRINT 630,D1,DC,PDX,ADX,D2C(2),D2CX(1),PDXX,ADXX,D2CX(2),FN2(1),
     1 FN2(2),Z
      GO TO 1020
  870 DO 950 I=1,20
      X3=(X1+X2)/2.
      FN2(1)= X3
      MX= 4
      GO TO 420
  880 ER3= D2C(2)
      IF(ABSF(PDX)-ABSF(ERP))890,890,900
  890 IF(ABSF(ADX)-ABSF(ERA))1170,1170,900
  900 IF(ER1)910,910,920
  910 IF(ER3)930,930,940
  920 IF(ER3)940,940,930
  930 X1= X3
```

```
      ER1= ER3
      GO TO 950
 940  X2= X3
      ER2=ER3
 950  CONTINUE
 960  PRINT 970
 970  FORMAT(24H0NO SOLUTION IN 20 TRIES)
      GO TO 1030
1020  MX= 5
      FN2(1)= X2
      GO TO 420
1030  PRINT 630,D1,DC,PDX,ADX,D2C(2),D2CX(1),PDXX,ADXX,D2CX(2),FN2(1),
     1 FN2(2),Z
1031  FN2X(2)=FN2X(2)+DFN2(2)
      IF(FN2F(2)-FN2X(2))50,50,815
1050  IF(QWP)1060,1070,1060
1060  MY=2
      GO TO 1080
1070  MY=1
1080  IF(FN-1.)1090,320,1090
1090  MY=3
      GO TO 320
1100  P=Z1
      A=Z2
      IF(QWP)1110,1120,1110
1110  XL=ACOSF(ABSF(COSDF(QWP)*COSDF(2.*P)))/2.
      DEL=ATN1F(SINDF(QWP),-SINDF(2.*P)/COSDF(2.*P))
      A=ATN1F(SINDF(A)/COSDF(A),COSF(XL)/SINF(XL))*57.2956
      AX=ATN1F(SINDF(Z3)/COSDF(Z3),SINF(XL)/COSF(XL))*57.2956
      PRINT 1115,A,AX
1115  FORMAT(14H0CALCULATED A=F15.4,4H ANDF15.4)
      P=DEL*28.6478-45.
      A=(A+AX)/2.
1120  IF(FN-1.)1130,1160,1130
1130  TNA=SINDF(A)/COSDF(A)
      IF (TNA) 1140,1150,1150
1140  TNA=-TNA
      P=P+90.
1150  P=(P+45.)/FN-45.
      A=ATANDF(TNA**(1./FN))
1160  GO TO (550,550,380,810,560),M
1170  IF(NCH)1175,1030,1175
1175  PRINT 630,D1,DC,PDX,ADX,D2C(2),D2CX(1),PDXX,ADXX,D2CX(2),FN2(1),
     1 FN2(2),Z
      NPCH= XABSF(NCH)
      DO 2011 I=1,NPCH
1180  FN2(1)=FN2(1)-DFN2(1)
      MX=6
      GO TO 420
1190  IF(ABSF(PDX)-EP)2000,2000,2015
2000  IF(ABSF(ADX)-EA)2010,2010,2015
2010  IF(FN2(1)-FN2I(1))2015,2015,2011
2011  CONTINUE
2015  PRINT 630,D1,DC,PDX,ADX,D2C(2),D2CX(1),PDXX,ADXX,D2CX(2),FN2(1),
     1 FN2(2),Z
```

```
      FN2(1)=X3
      DO 2060 I=1,NPCH
 2020 FN2(1)=FN2(1)+DFN2(1)
      MX=7
      GO TO 420
 2030 IF(ABSF(PDX)-EP)2040,2040,1030
 2040 IF(ABSF(ADX)-EA)2050,2050,1030
 2050 IF(FN2(1)-FN2F(1))2060,2060,1030
 2060 CONTINUE
      GO TO 1030
 2070 EP= Z1
      EA =Z2
      GO TO 50
       END
      SUBROUTINE CS(C3,X,FN3)
      DIMENSION C3(2),FN3(2),T1(2),T2(2),T3(2)
      T1(1)=X
      T1(2)=0.
      CALL CD(T2,T1,FN3)
      CALL CM(T3,T2,T2)
      T1(1)=1.-T3(1)
      T1(2)=-T3(2)
      CALL CSR(C3,T1)
C     X=N1*SIN(PHI1)
      RETURN
      END
      SUBROUTINE SRC(R,T1,T2)
      DIMENSION R(2),T1(2),T2(2),T3(2),T4(2)
      DO 1 I=1,2
      T3(I)=T1(I)-T2(I)
    1 T4(I)=T1(I)+T2(I)
      CALL CD(R,T3,T4)
      RETURN
      END
      SUBROUTINE RC(CR,RP21,RP32,Q)
C     MCCRACKIN   REFLECTION COEFFICIENT OF A FILM  Q=N2*COS(PHI2)*D/WL
      DIMENSION CR(2),RP21(2),RP32(2),Q(2),T1(2),T2(2),T3(2),P(2)
      P(1)=12.56637*Q(1)
      P(2)=12.56637*Q(2)
      T2(1)=EXPF(P(2))*COSF(P(1))
      T2(2)=-EXPF(P(2))*SINF(P(1))
      CALL CM(T1,T2,RP32)
      CALL CM(T3,T1,RP21)
      T1(1)=RP21(1)+T1(1)
      T1(2)=RP21(2)+T1(2)
      T3(1)=1.+T3(1)
      CALL CD(CR,T1,T3)
      RETURN
      END
      SUBROUTINE CSR(SRX,X)
      DIMENSION SRX(2),X(2)
C     SRX=COMPLEX SQRT X,IF SRX(2)=0,SRX(1)=POSITIVE,IF SRX(1)=0,
C     SRX(2)=NEGATIVE,OTERWISE SRX(1)=POSITIVE AND SRX(2)=CORRECT SIGN
      IF(X(2))4,1,4
    1 IF(X(1))2,2,3
```

```
2 SRX(1)=0.
  SRX(2)=-SQRTF(ABSF(X(1))
  GO TO 5
3 SRX(1)=SQRTF(X(1))
  SRX(2)=0.
  GO TO 5
4 R=SQRTF(X(1)**2+X(2)**2)
  SRX(1)=SQRTF((R+X(1))/2.)
  SRX(2)=SIGNF(SQRTF((R-X(1))/2.),X(2))
5 RETURN
  END
  SUBROUTINE  CD(T2,T1,T3)
  DIMENSION T1(2),T2(2),T3(2)
  E=T3(1)**2+T3(2)* *2
  T2(1)=(T1(1) *T3(1)+ T1(2)*T3(2))/E
  T2(2)=(T1(2)*T3(1)-T1(1)*T3(2))/E
  RETURN
  END
  SUBROUTINE CM(T3,T1,T2)
  DIMENSION T1(2),T2(2),T3(2)
  T3(1)=T1(1)*T2(1)-T1(2)*T2(2)
  T3(2)=T1(1)*T2(2)+T1(2)*T2(1)
  RETURN
  END
```

7. References

[1] O. Heavens, Optical Properties of Thin Solid Films (1955) Butterworth Scientific Publications, London

[2] A. B. Winterbottom, Optical Studies of Metal Surfaces, The Royal Norwegian Scientific Society Report No. 1, (1955) published by F. Bruns, Trondheim, Norway

[3] F. L. McCrackin, E. Passaglia, R. R. Stromberg and H. L. Steinberg, Measurement of the Thickness and Refractive Index of Very Thin Films and the Optical Properties of Surfaces by Ellipsometry, J. Research NBS A 67, 363, July-August 1963

[4] Proceedings of the Symposium on the Ellipsometer and its Use in the Measurement of Surfaces and Thin Films, National Bureau of Standards, Sept. 5, 6, 1963, to be published by NBS, for sale by the Supt. of Documents, U.S. Government Printing Office, Washington, D. C.

[5] R. R. Stromberg, E. Passaglia and D. Tutas, J. Research NBS 67A, (Phys. and Chem.) 431 (1963)



Fig. 1. Schematic diagram of 5 films on a substrate. Refractive indices of films are n_2 to n_6, thicknesses of films are d_2 to d_6, angle of incidence is ϕ, and refractive indices of surrounding medium and substrate are n_1 and n_7.



Fig. 2. Flow Diagram of Computer Program. Numbers in ovals refer to statement numbers in fortran program. A, B, C, D, D1, and D2 in squares refer to detail diagrams shown in the following figures.



Fig. 3. Diagrams A and B. A: Determines MY and stores range of values for film thickness for instruction no. 902. B: Corrects ellipsometer readings for retardation of compensator (QWP) and number of reflections (FN).



Fig. 4. Diagram C. Calculates and prints ellipsometer readings (PX and Ax or PXX, AP and AS) and reflection coefficients (RP_{21} and RN_{21}).



Fig. 5. Diagram D. Interprets NCH. If NCH is zero, no calculation is performed, if NCH is not zero, the limits of error for FN2(1) are determined and printed.

* U.S. GOVERNMENT PRINTING OFFICE : 1964 O—730-636

DEPARTMENT OF COMMERCE
NATIONAL BUREAU OF STANDARDS